# delimit ;
clear ;
set more off ;
est drop _all ;
cd "R:/personlig/fenellac/menarche_replication/analysis/" ;

global plusdir "R:/personlig/fenellac/stata_ado/plus/" ;
sysdir set PLUS $plusdir ;
adopath ++ $plusdir ;

* **************************************************************************** ;
* This code replicates Appendix G, Table 1.
*
* Paper: "Age of Marriage and Women's Political Engagement: Evidence from India"
* Authors: Fenella Carpena, Francesca Jensenius
* 
* Code by Fenella Carpena
* Last update: March 3, 2020
* **************************************************************************** ;

****************************************************************************** ;
* select the sample
****************************************************************************** ;

use "./input/eligible-women-ihds02-lhs.dta", clear ;
merge 1:1 IDHH PERSONID using "./input/eligible-women-ihds02-rhs.dta" ;
tab _merge ;
assert _merge == 3 ;
drop _merge ;

* keep only rural women ;
keep if rural == 1 ;

* keep only women who are married only once (i.e., have not ever remarried) ;
drop if ever_remarried == 1 ;
  
* keep only women age of menarche between 11-18 (1st-99th pctile) ; 
summ menarche_age ;
keep if menarche_age >= 11 & menarche_age <= 18 ;

* generating district identifier ;
egen dt_id = group(STATEID DISTID) ;

****************************************************************************** ;
* label variables for the output table 
****************************************************************************** ;

label var ever_work "\shortstack[l]{Ever worked for pay}" ;
label var wagework_any "\shortstack[l]{Worked for pay last year}" ;
label var current_work "\shortstack[l]{Currently working for pay}" ;

local employment "ever_work wagework_any current_work" ;

label var marriage_age "Marriage Age" ;

****************************************************************************** ;
* TABLE: IV regressions on employment
****************************************************************************** ;
est drop _all ;

foreach var of varlist `employment' { ;
 	ivreghdfe `var' height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ (marriage_age = menarche_age), a(dt_id) robust ;
	est sto `var' ;
	estadd local distFE "Yes" ;
} ;

esttab * using "./output/table-IV-employment.tex",
	replace msign(--)
	drop(height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ) 
	cells(b(label() star fmt(%9.3f %9.3f)) se(par)) 
	star(* 0.10 ** 0.05 *** 0.01) 
	stats(N, fmt(%9.0f) labels("Observations"))
	prehead(\begin{table}[h]\centering \vspace{-1.5ex} \normalsize `"\def\sym#1{\ifmmode^{#1}\else\(^{#1}\)\fi}"'
		\captionsetup{justification=centering} 
		\caption{Effects on Employment, Rural Women}
		\label{table-IV-employment}
		\begin{tabular*}{\hsize}{@{\hskip\tabcolsep\extracolsep\fill}lp{2cm}p{4cm}p{4cm}}
		\toprule)
	legend label  booktabs  collabels( , none)
	postfoot(`"\bottomrule"'  \end{tabular*} \captionsetup{justification=justified, width=\hsize} 
		\caption*{\small \textit{Notes:} 
		IV regression, \textit{Marriage Age} instrumented with \textit{Menarche Age}.
		Data from IHDS 2011-12, rural women.
		All regressions include district FEs, height, age, parents' education, caste, religion. 
		Robust SEs in parenthesis. ***\$\,p < 0.01$, **\$\,p < 0.05$, *\$\,p<0.10$.\looseness=-1}
		\end{table}) ;

exit ;

